/* Everyting inside here is comment. */
/* The SAS interface has three windowns: programer editor, output and log.

You need to submit code from the editor window to execute it (by click menu "run" and 

then "submit", or use the shortcut in the toolbar). */
/* Dark and light blue colors  indicate SAS key words. Texts in black are user-defined. */
/* Create SAS data sets */
data a;

input id survtime cen age gender $;

datalines;

1 3.4 1 45 M

2 4.6 0 67 F

.

.

.

;

run;

/* work on the data */
data a;

set a;

sex=(gender="M");

run;

/* Or */
data a;

infile "c:\mydata\lungcancer.dat"; 

input id survtime cen age gender;

run;

/* data created as above are temporary. To make them permanent, we need to create 

a data library, and save data in it. */
libname abc "c:\mydata";

data abc.lung;

infile "c:\mydata\lungcancer.dat"; 

input id survtime cen age gender;

run;

/* Or make a temporary one permanent */
data abc.lung;

set a;

run;

/* to see the data, proc print will put them in the "output" window. */
proc print data=abc.lung;

run;

/* to see the variables type (char or num) */
Proc contents data=abc.lung;

run;

/* fitting Cox proportional hazards (PH) models */
proc phreg data=abc.lung;

model survtime*cen(0)=age sex;

run;

/* The value in () is the censoring indicator. 

Pay attention to the data to make sure the right value is used here. 

In most data sets, 1=failure and 0=censoring, but not always. */
/* go back to Kaplan-Meier plots */
proc lifetest data=abc.lung plot=(s);

time survtime*cen(0);

run;

proc lifetest data=abc.lung plot=(s);

time survtime*cen(0);

strata gender;

run;

/* Plot the cumulative hazard function, which is -log(S(t)) */
proc lifetest data=abc.lung plot=(ls);

time survtime*cen(0);

run;

/* Return to Cox model */
/* Suppopse this is data set b

ID Eventtime  Status  Cov(time=5)  Cov(time=8)  Cov(time=10)   Cov(time=12)

1       5       1               2

2       8       0(censored)     2.3       4.6

3       10      1               2.1       4.2              6.1

4       12      1               2.9       4.8              6.7           8

....

*/
/* And this is data set c (counting process data format)

ID     start   end     eventtime       status  cov

1       0       5       5               1       2

2       0       5       8               0       2.3

2       5       8       8               0       4.6

3       0       5       10              0(Note) 2.1

3       5       8       10              0(Note) 4.2

3       8       10      10              1       6.1

4       0       5       12              0(Note) 2.9

4       5       8       12              0(Note) 4.8

4       8       10      12              0(note) 6.7

4       10      12      12              1       8

;

.

.

.

Note: At time 5, subject 2 has not died yet, so the status is 0. 

Similarly for the other 0's followed by a "(Note)". For censored subjects, 

status is always 0. For each died subject, there is only one "status=1" 

at the end (its own death time).*/
/* Then SAS can fit data c by the following. */
proc phreg data=c;

model end*status(0)=cov /entry=start;

run;

/* "entry" in the above should be blue, but SAS forgets it.

This does happen for some key words */ 

/* Another data set d. Want to include x and x*log(t) in the model.

ID Eventtime  Status            X

1       5       1               2

2       8       0(censored)     2.3       

3       10      1               2.1       

4       12      1               2.9  

*/
proc phreg data=d;

model eventtime*status(0)=x xt;

xt=x*log(eventtime);

run;

/* "xt=x*log(t)" in the above is called a programming statement. 

It will create dynamic random variable, i.e., it changes value at each eventtime.

It can NOT be moved to a DATA step, where all variables are static. */
/* Now abuse the above data a little bit. Suppose x is actually the kidney tranplant

time for a patient. We would like to fit a Cox model with transplant as a time-dependent 

covariate. */
proc phreg data=d;

model eventtime*status(0)=transplant;

if eventtime < x then transplant=0;

else transplant=1;

run;

/* Note the explanatory variable "transplant" is not even in the data set d (similarly for the variable xt in the previous example).  It is created 

when SAS executes the above proc phreg. After it is done, the data set d still does not have

a variable called "transplant". */
/* If you try to use R to analyze the transplant effect by the above model, you have to manually 

create a data set satisfying "if eventtime < x then transplant=0; else transplant=1;". And this 

new data set can be huge. Its sample size is roughly n*n/2, where n is the original sample size. 

See the handout "Using R to fit Cox Proportional Hazards Model with time-dependent covariates"

to figure out how to create the new data set if you want to. */
